Melting Using Element Death

Introduction

This tutorial was completed using ANSYS 7.0 The purpose of the tutorial is to outline the steps required to use element death to model melting of a material. Element death is the "turning off" of elements according to some desired criterion. The elements are still technically there, they just have zero stiffness and thus have no affect on the model.

This tutorial doesn't take into account heat of fusion or changes in thermal properties over temperature ranges, rather it is concerned with the element death procedure. More accurate models using element death can then be created as required. Element birth is also possible, but will not be discussed here. For further information, see Chapter 10 of the Advanced Guide in the ANSYS help file regarding element birth and death.

The model will be an infinitely long rectangular block of material 3cm X 3cm as shown below. It will be subject to convection heating which will cause the block to "melt".



Preprocessing: Defining the Problem

1. **Give example a Title**

   Utility Menu > File > Change Title ...
   `/title, Element Death`
2. **Open preprocessor menu**

   ANSYS Main Menu > Preprocessor
   `/PREP7`
3. **Create Rectangle**

   Preprocessor > Modeling > Create > Areas > Rectangle > By 2 Corners

   Fill in the window with the following dimensions:
   WP X = 0
   WP Y = 0
   Width = 0.03
   Height = 0.03
   `BLC4,0,0,0.03,0.03`
4. **Define the Type of Element**

   Preprocessor > Element Type > Add/Edit/Delete...

   For this example, we will use PLANE55 (Thermal Solid, Quad 4node 55). This element has 4 nodes and a single DOF (temperature) at each node. PLANE55 can only be used for 2 dimensional steady-state or transient thermal analysis.
5. **Define Element Material Properties**

   Preprocessor > Material Props > Material Models > Thermal > Conductivity > Isotropic
   In the window that appears, enter the following properties:
   i. Thermal Conductivity KXX: 1.8

   Preprocessor > Material Props > Material Models > Thermal > Specific Heat

In the window that appears, enter the following properties:

ii. Specific Heat C: 2040

Preprocessor > Material Props > Material Models > Thermal > Density

In the window that appears, enter the following properties:

iii. Density DENS: 920

6. **Define Mesh Size**

   Preprocessor > Meshing > Size Cntrls > ManualSize > Areas > All Areas...

   For this example we will use an element edge length of 0.0005m.

7. **Mesh the frame**

   Preprocessor > Meshing > Mesh > Areas > Free > click 'Pick All'

Solution Phase: Assigning Loads and Solving

1. **Define Analysis Type**

   Solution > Analysis Type > New Analysis > Transient

   The window shown below will pop up. We will use the defaults, so click OK.



   ANTYPE,4

2. **Turn on Newton-Raphson solver**

   Due to a glitch in the ANSYS software, there is no apparent way to do this with the graphical user interface. Therefore, you must type **NROPT,FULL** into the commmand line. This step is necessary as element killing can only be done when the N-R solver has been used.

3. **Set Solution Controls**

   Solution > Analysis Type > Sol'n Controls

   The following window will pop up.



A) Set `Time at end of loadstep` to 60 and `Automatic time stepping` to OFF.
B) Set `Number of substeps` to 20.
C) Set the `Frequency` to Write every substep.
Click on the NonLinear tab at the top and fill it in as shown



D) Set `Line search` to ON .
E) Set the `Maximum number of iterations` to 100.

For a complete description of what these options do, refer to the help file. Basically, the time at the end of the load step is how long the transient analysis will run and the number of substeps defines how the load is broken up. By writing the data at every step, you can create animations over time and the other options help the problem converge quickly.

4. **Apply Initial Conditions**

Solution > Define Loads > Apply > Initial Condit'n > Define > Pick All

Fill in the IC window as follows to set the initial temperature of the material to 268 K:



5. **Apply Boundary Conditions**

   For thermal problems, constraints can be in the form of Temperature, Heat Flow, Convection, Heat Flux, Heat Generation, or Radiation. In this example, all external surfaces of the material will be subject to convection with a coefficient of 10 W/m^2*K and a surrounding temperature of 368 K.

   Solution > Define Loads > Apply > Thermal > Convection > On Lines > Pick All

   Fill in the pop-up window as follows, with a film coefficient of 10 and a bulk temperature of 368.



The model should now look as follows:



6. **Solve the System**

   Solution > Solve > Current LS
   `SOLVE`

Postprocessing: Prepare for Element Death

1. **Read Results**

   General Postproc > Read Results > Last Set
   `SET,LAST`

2. **Create Element Table**

   Element death can be used in various ways. For instance, the user can manually kill, or turn off, elements to create the desired effect. Here, we will use data from the analysis to kill the necessary elements to model melting. Assume the material melts at 273 K. We must create an element table containing the temperature of all the elements.

   - From the **General Postprocessor** menu select **Element Table > Define Table...**
   - Click on 'Add...'
   - Fill the window in as shown below, with a title `Melty` and select DOF solution > Temperature TEMP and click OK.



   We can now select elements from this table in the temperature range we desire.

3. **Select Elements to Kill**

   Assume that the melting temperature is 273 K, thus any element with a temperature of 273 or greater must be killed to simulate melting.

   Utility Menu > Select > Entities
   Use the scroll down menus to select Elements > By Results > From Full and click OK.



Ensure the element table `Melty` is selected and enter a VMIN value of 273 as shown.



Solution Phase: Killing Elements

1. **Restart the Analysis**

   Solution > Analysis Type > Restart > OK

   You will likely have two messages pop up at this point. Click OK to restart the analysis, and close the warning message. The reason for the warning is ANSYS defaults to a multi-frame restart, which this analysis doesn't call for, thus it is just warning the user.

2. **Kill Elements**

   The easiest way to do this is to type **ekill,all** into the command line. Since all elements above melting temperature had been selected, this will kill only those elements.

   The other option is to use **Solution > Load Step Opts > Other > Birth & Death > Kill Elements** and graphically pick all the melted elements. This is much too time consuming in this case.

Postprocessing: Viewing Results

1. **Select Live Elements**

   Utility Menu > Select > Entities

Fill in the window as shown with **Elements > Live Elem's > Unselect** and click Sele All.



With the window still open, select **Elements > Live Elem's > From Full** and click OK.



2. **View Results**

General Postproc > Plot Results > Contour Plot > Nodal Solu > DOF solution > Temperature TEMP

The final melted shape should look as follows:



This procedure can be programmed in a loop, using command line code, to more accurately model element death over time. Rather than running the analysis for a time of 60 and killing any elements above melting temperature at the end, a check can be done after each substep to see if any elements are above the specified temperature and be killed at that point. That way, the prescribed convection can then act on the elements below those killed, more accurately modelling the heating process.

Command File Mode of Solution

The above example was solved using a mixture of the Graphical User Interface (or GUI) and the command language interface of ANSYS. This problem has also been solved using the ANSYS command language interface that you may want to browse. Copy and paste following code into Notepad or a similar text editor and save it to your computer. Now go to **'File > Read input from...'** and select the file.

```
finish
/clear

/title, Convection Example
/prep7                        ! Enter the preprocessor

! define geometry

k,1,0,0                       ! Define keypoints
k,2,0.03,0
k,3,0.03,0.03
k,4,0,0.03
a,1,2,3,4                     ! Connect the keypoints to form area


! mesh 2D areas

ET,1,Plane55                  ! Element type

MP,Dens,1,920                 ! Define density
mp,c,1,2040                   ! Define specific heat
mp,kxx,1,1.8                  ! Define heat transfer coefficient
```

```
esize,0.0005                ! Mesh size
amesh,all                   ! Mesh area

finish
/solu                       ! Enter solution phase

antype,4                    ! Transient analysis

time,60                     ! Time at end of analysis

nropt,full                  ! Newton Raphson - full
lumpm,0                     ! Lumped mass off
nsubst,20                   ! Number of substeps, 20
neqit,100                   ! Max no. of iterations
autots,off                  ! Auto time search off
lnsrch,on                   ! Line search on
outres,all,all              ! Output data for all substeps
kbc,1                       ! Load applied in steps, not ramped

IC,all,temp,268             ! Initial conditions, temp = 268

nsel,s,ext                  ! Node select all exterior nodes
sf,all,conv,10,368          ! Apply a convection BC
nsel,all                    ! Reselect all nodes
/gst,off                    ! Turn off graphical convergence monitor

solve
finish

/post1                      ! Enter postprocessor
set,last                    ! Read in last subset of data
etable,melty,temp,          ! Create an element table
esel,s,etab,melty,273       ! Select all elements from table above 273
finish

/solu                       ! Re-enter solution phase
antype,,rest                ! Restart analysis
ekill,all                   ! Kill all selected elements
esel,all                    ! Re-select all elements

finish

/post1                      ! Re-enter postprocessor
set,last                    ! Read in last subset of data
esel,s,live                 ! Select all live elements
plnsol,temp                 ! Plot the temp contour of the live elements
```

Reference

http://www.mece.ualberta.ca/tutorials/ansys

Web: www.AnsysHelp.ir
Email: AnsysHelp.ir@gmail.com